On the Veri�cation of VDM Speci�cation and Re�nement with PVS

Savi Maharaj� Department of Computing and Mathematics Stirling University Stirling FK9 4LA, UK savi@cs.stir.ac.uk

Juan Bicarregui Rutherford Appleton Laboratory and Imperial College, London

Abstract

Although the formal method VDM has been in existence since the 1970’s, there are still no satisfactory tools to supŋ port veri^cation in VDM. This paper deals with one posŋ sible means of approaching this problem by using the PVS theoremŋprover. It describes a translation of a VDMŋSL speci^cation into the PVS speci^cation language using, esŋ sentially, the very transparent translation methods described in [1]. PVS was used to typecheck the speci^cation and to prove some nonŋtrivial validation conditions. Next, a more abstract speci^cation of the same system was also expressed in PVS, and the original speci^cation was shown to be a re^nement of this one. The drawbacks of the translation are that it must be done manually (though automation may be possible), and that the Şshallow embeddingŤ technique which is used does not accurately capture the proof rules of VDMŋSL. The bene^ts come from the facts that the portion of VDMŋSL which can be represented is substantial and that it is a great advantage to be able to use the powerful PVS proofŋchecker.

1. Introduction

In the paper [1], Sten Agerholm demonstrates how a large subset of VDMŋSL[4] may be translated directly into the speci^cation language of the PVS theorem prover[8]. The translation is a Şshallow embeddingŤ which works well in this case because the two languages possess many structures and concepts in common. The result is a simple and direct means of harnessing the power of the PVS proofŋchecker for verifying results about VDMŋSL speci^cations.

We independently discovered essentially the same repreŋ sentation of VDMŋSL in PVS, and explored this by transŋ lating a medium sized VDMŋSL speci^cation into PVS and

�Savi Maharaj would like to thank the European Research Consortium in Informatics and Mathematics for supporting this work.

proving several theorems about it. Next, we considered VDMŋstyle re^nement, and found that the PVS theory mechŋ anism can be used to express this quite satisfactorily. Again, we used our example to explore this approach, and found that we were able to prove all of the re^nement proof obliŋ gations with the PVS proofŋchecker.

The VDMŋSL example which is used was developed with the help of the IFAD VDMŋSL Toolbox [9] which proŋ vides syntaxŋchecking and some typechecking for VDMŋ SL. Other work which deals with re^nement within PVS includes [12] which is concerned with providing proof supŋ port for Abrial’s ŞBŤ Abstract Machine Notation.

The main contributions of this paper are a mediumŋsized case study with nonŋtrivial theorems proved giving evidence to support the feasibility of the approach proposed in [1] and a simple extension of this approach to incorporate reŋ ^nement. The paper is organized as follows: Section 2 describes the application which serves as our main examŋ ple. In Section 3 we show how this system was formally speci^ed in VDMŋSL, the speci^cation translated into PVS, and a number of proof obligations and validation conditions formally veri^ed using PVS. In Section 4 we present our ^rst speci^cation as a re^nement of a more abstract speci^ŋ cation, and discuss using PVS to prove the re^nement proof obligations. In the ^nal sections, we present some obserŋ vations about what was learned during this project and our conclusions.

2. The MSMIE protocol

MSMIE (Multiprocessor SharedŋMemory Information Exchange) is a protocol for Şinterŋprocessor communicaŋ tions in distributed, microprocessorŋbased nuclear safety systemsŤ [13], which has been used in the embedded softŋ ware of Westinghouse nuclear systems designs.

The protocol uses multiple buffering to ensure that no ŞdataŋtearingŤ occurs as separate processors communicate via some shared memory. In other words, data should never be overwritten by one process while it is still being read by another. One important requirement is that neither writing nor reading processes should have to wait for a buffer to become available; another is that recent information should be passed, via the buffers, from writers to readers. The exŋ ample has previously been analyzed using CCS in [5] and is used in as a basis of a comparison of VDM and B in [7]. In a simpli^cation considered in [5] it is assumed that information is being passed from a single ŞslaveŤ procesŋ sor to several ŞmasterŤ processors. Thus, there are several reading processors, ŞmastersŤ, but only one writing, ŞslaveŤ process.

The information exchange is realized by a system with three buffers. At any time, one buffer is available for writing, one for reading, and the third is either between a write and a read and hence contains the most recently written information, or between a read and a write and so is idle.

The status of each buffer is recorded by a ~ag which can take one of four values:

s ŋ Şassigned to slaveŤ This buffer is reserved for writing.

It may actually be being written at the moment or just marked as available for writing.

n ŋ ŞnewestŤ This buffer has just been written and contains the latest information. It is not being read at the moment.

m ŋ Şassigned to masterŤ This buffer is being read by one or more processors.

i ŋ ŞidleŤ This buffer is not being read or written and does not contain the latest data.

The names of the master processors that are currently reading are also stored in the state.

The VDM speci^cation of [6] and [7] is concerned with various Şdata modelsŤ of MSMIE: the state of the device is modeled but not the slave and master processors nor the dynamic evolution of the system as they access the buffers in parallel. This analysis concerns only the operations which modify the buffer status ~ags. In the system as a whole, these operations are protected by a system of semaphores which allows each operation uninterrupted access to the state, and thus their behavior is purely sequential.

There are three such operations:

slave This operation is executed when a write ^nishes. It sets the status of the buffer that was being written to ŞnewestŤ, thereby replacing any other buffer with this status.

acquire This is executed when a read begins. The new reader name (passed as a parameter) is added to the set of readers and status ~ags are updated as appropriate.

release This is executed when a read ends. The given reader name is removed from the set of readers and status ~ags are updated as appropriate.

The precise description of these operations is left to the formal speci^cation in the following section.

We note, in passing, that the MSMIE protocol has the undesirable property that it is possible for information ~ow from slave to master to be held up inde^nitely. This problem is dealt with in the original paper [13] by the use of timing constraints. The paper [5] suggests an improvement to the protocol in which the problem is avoided by the use of a fourth buffer. This improved protocol is quite intricate, and is modeled in [6] by using nonŋstandard extensions to VDM which provide a more concise means of expression than standard VDM. In this paper we restrict ourselves to standard VDM speci^cations and do not treat the improved MSMIE protocol.

3. Formal speci�cations of MSMIE

The paper [6] explores several ways of specifying this system in VDM with varying degrees of abstraction. These speci^cations can be translated more or less Şas isŤ into PVS, which can then be used to carry out proof obligations. The translation can be done in such a way that some of the proof obligations are automatically generated by PVS, though it is still necessary to type in others by hand.

The VDM speci^cations shown here have been modi^ed to conform with standard VDMŋSL so they differ slightly from those in [6].

3.1. The VDM speci�cation

The speci^cation uses an auxiliary function called count, which counts the number of occurrences of a given item in a sequence.

functions

count[@T ] : @T × @T � → N

count (s, ss) 4

cases ss :

[] → 0, others → if hd ss = s then 1 + count[@T ] (s, tl ss) else count[@T ] (s, tl ss) end The possible values of the status ~ags are given via an enumerated type; the type of the names of master processes is deferred.

types

Status = S | M | N | I;

MName = token

The state records the status of each of the three buffers, and the names of all currently reading master processes. These are represented by, respectively, a sequence of status values and a set of master names. The invariant captures constraints on the possible states that are reachable: there is exactly one buffer assigned to the writing slave process; at most one buffer is currently being read, and at most one holds newest data that is not being read; the set of readers is empty precisely when no buffer is being read. In the initial state, one buffer is assigned to the slave and the other two buffers are marked as idle.

state Σ of

b : Status� ms : MNameŋset

inv mkŋΣ (b, ms) 4

len b = 3 ∧ count[Status] (S, b) = 1 ∧ count[Status] (M, b) ∈ {0, 1} ∧ count[Status] (N, b) ∈ {0, 1} ∧ (count[Status] (M, b) = 0 ⇔ ms = {})

init s 4 s = mkŋΣ ([S, I, I], {}) end The slave operation is executed when a slave process completes writing. The buffer that has just been written, previously of status S, is given the status N re~ecting that it now contains the newest data. This buffer replaces any other buffer with the N status. The operation also selects one of the available buffers and assigns to it status S, making it the new buffer available for writing.

operations

slave () ext wr b : (Status�) pre true post ∀ i ∈ {1, 2, 3} · ( ↼− b (i) = S ⇒ b (i) = N) ∧ ( ↼− b (i) = M ⇒ b (i) = M) ;

The acquire operation is executed when a master process is about to begin reading. The new reader’s name, passed as a parameter, is added to the set of active readers, and status ~ags are updated as necessary. If there is already a buffer being read, then the new reader also begins to read that buffer and no status changes are needed. Otherwise it reads the buffer with the newest data, status N, and the status of that buffer is changed to M.

acq (l : MName) ext wr b : (Status�) wr ms : (MNameŋset) pre (¬ (l ∈ ms)) ∧ (∃ i ∈ {1, 2, 3} · b (i) = N ∨ b (i) = M)

post ms = ↼−ms ∪ {l} ∧ ∀ i ∈ {1, 2, 3} · if ↼− b (i) = N ∧ ↼−ms = {} then b (i) = M

else b (i) = ↼− b (i) ;

The release operation takes place when a master process has ^nished reading. Because of space limitations in this paper, the speci^cation of this operation is not shown.

3.2. The PVS system

The PVS system[8] was developed at SRI International in California. It features an expressive speci^cation language based on classical higherŋorder logic, enhanced with a rich system of types including subtyping, dependent types and recursive types. Speci^cations may be structured using a system of parameterized theories. The PVS speci^cation language is supported by a collection of tools including a type checker and an interactive proof checker in which basic proof commands are tightly integrated with powerful decision procedures and tactics. The main attraction of the method exempli^ed in this paper is that it enables us to use this powerful proofŋchecker for veri^cation of VDM.

3.3. PVS translation

The VDM speci^cation is represented as a PVS theory called msmie sigma2. The theory is parameterized over a nonŋempty type of master names.

msmie_sigma2[MName : TYPE+] : THEORY

  BEGIN

The theory begins by importing another theory containing de^nitions of functions on lists. The ŞcountŤ function is de^ned within this imported theory.

IMPORTING list_funs

As in VDM, the possible values of the status ~ags are represented by an enumerated type.

Status : TYPE = {Slave,Master,Newest,Idle}

The state and invariant of the VDM speci^cation are repŋ resented by a single (dependent) type in PVS. The de^nition is best understood in two parts. First, the state is repreŋ sented as a record containing two ^elds: a list, b of Status values, and a set of master names, ms. The set of valid states is then represented by forming the subtype, sigma2, of all such records which satisfy the invariant.

sigma2 : TYPE =
 {x : [# b : list[Status],
  ms : setof[MName] #] |
 ((length(b(x)) = 3) AND (count(Slave,b(x)) = 1)
  AND
  member(count(Master,b(x)),{x:nat|x=0 OR x=1})
  AND
  member(count(Newest,b(x)),{x:nat|x=0 OR x=1})
  AND
  ((count(Master,b(x)) = 0) <=>
   (ms(x) = emptyset)))}

The initial state is de^ned as a record containing apŋ propriate values and its type is explicitly constrained to be

sigma2. When this de^nition is typechecked, PVS will auŋ

tomatically generate as a typeŋchecking constraint (TCC) the condition that the initial state satis^es the invariant. We shall look more closely at these TCCs in the next section.

initial_sigma2 : sigma2 =
  (# b := (: Slave, Idle, Idle :),
     ms := emptyset #)

For each operation, the precondition is represented as a predicate over the valid states (sigma2). The postcondition is represented as a predicate over pairs of valid states. The operation is then de^ned as a function which maps each value from the subtype de^ned by the precondition to a nondeterministically chosen valid state which, when paired with the input value, satis^es the postcondition^1.

Typechecking such a de^nition causes PVS to generate a TCC stating that a suitable value for the nonŋdeterministic choice does indeed exist. In other words, we are asked to show that the speci^ed operation is feasible.

Our ^rst example is the slave operation. The de^nitions of the preŋ and postconditions resemble very closely the original VDM speci^cation. One difference is that there is no analogue in PVS to the frames in VDM, so that the fact that the variable ms has readŋonly status in the VDM speciŋ ^cation must be explicitly stated in the PVS postcondition.

pre_slave:[sigma2 ŋ> bool] = LAMBDA (st:sigma2) :
  true

post_slave:[(pre_slave),sigma2 ŋ> bool] =
  LAMBDA (st:(pre_slave),st2:sigma2) :

1The interpretation of looseness in VDM and PVS is discussed in Secŋ tion 5.2.

((FORALL (i:{x:nat| x=0 OR x=1 OR x=2}) :
   (nth(b(st),i) = Slave IMPLIES
    nth(b(st2),i) = Newest)
   AND
   (nth(b(st),i) = Master IMPLIES
    nth(b(st2),i) = Master))
 AND
 (ms(st2) = ms(st)))

The slave operation is then de^ned using the nondeterŋ ministic choice operator ŞchooseŤ.

slave:[(pre_slave) ŋ> sigma2] =
  LAMBDA (st:(pre_slave)) :
    choose({st2:sigma2 | post_slave(st,st2)})

The speci^cations of acquire and release are done simiŋ larly. First the precondition and postcondition are de^ned as predicates.

pre_acq:[MName ŋ>[sigma2 ŋ> bool]] =
  LAMBDA (l:MName)(st:sigma2) :
    (NOT member(l,ms(st))) AND
    (EXISTS (i:{x:nat| x=0 OR x=1 OR x=2}) :
       (nth(b(st),i) = Newest OR
        nth(b(st),i) = Master))

post_acq:
  [l:MName ŋ> [(pre_acq(l)),sigma2 ŋ> bool]] =
  LAMBDA (l:MName)(st:(pre_acq(l)),st2:sigma2) :
    (ms(st2) = union(ms(st),singleton(l))) AND
    (FORALL (i:{x:nat| x=0 OR x=1 OR x=2}) :
       IF nth(b(st),i) = Newest AND
          ms(st) = emptyset
       THEN nth(b(st2),i) = Master
       ELSE nth(b(st2),i) = nth(b(st),i)
       ENDIF)

Next, the acquire operation is de^ned as a function which, when given a master name l and a state belonging to the type (pre_acq(l)), returns a nondeterministically chosen state in sigma2 such that the two states together satisfy the postcondition post_acq(l).

acq:[l:MName, (pre_acq(l)) ŋ> sigma2] =
  LAMBDA (l:MName, st:(pre_acq(l))) :
    choose({st2:sigma2 | post_acq(l)(st,st2)})

The speci^cation of the release operation is not shown.

3.4. Typechecking constraints

Typechecking a PVS theory usually results in the creŋ ation of proof obligations called typechecking constraints (TCCs). These must be proved in order to demonstrate that the theory is wellŋtyped. Simple TCCs can be handled by invoking an automatic TCCŋprover, tcp, but more dif^cult ones must be proved interactively by the user. In the case of the theory msmie_sigma2, 4 of the TCCs which are genŋ erated are too dif^cult for tcp to prove. Interestingly, these TCCs correspond to the satis^ability proof obligations for the speci^cation: in effect, we must show that the initial state satis^es the invariant, and that each of the three operations is feasible.

3.4.1. Showing that the initial state satis�es the invariant

The ^rst TCC generated for msmie_sigma2 is shown below. It results from the fact that we have explicitly stated that the initial state, initial_sigma2, is of type sigma2. We are required to prove that initial_sigma2 satis^es the prediŋ cate which de^nes the subtype sigma2. In other words, we must show that the initial state satis^es the invariant. The proof is too dif^cult for tcp, but can be done quickly using the interactive prover. The main goal, which consists of a conjunction of 5 formulae, is split, making each conjunct into a separate subgoal. Each of these is then proved by reŋ peatedly expanding de^nitions until a statement is obtained which PVS recognizes to be trivially true.

initial_sigma2_TCC1: OBLIGATION
  ((length((: Slave, Idle, Idle :)) = 3) AND
   (count(Slave,(: Slave, Idle, Idle :)) = 1)
        AND
   member(count(Master, (: Slave, Idle, Idle :)),
          {x: nat | x = 0 OR x = 1})
        AND
   member(count(Newest, (: Slave, Idle, Idle :)),
          {x: nat | x = 0 OR x = 1})
        AND
   ((count(Master, (: Slave, Idle, Idle :)) = 0)
     <=> (emptyset = emptyset)));

3.4.2. Showing that the operations are feasible

To specify the operations we used the nondeterministic choice operator of PVS. For this to be correctly typed, PVS requires us to demonstrate that there exist possible candiŋ dates for the nondeterministically chosen values. In other words, we must show that each operation is feasible.

We show the statement of this proof obligation for the acŋ quire operation. Given any master name l, and any state st within the type de^ned by the precondition, (pre_acq(l)), we must prove that the set of all states, st2, satisfying the postcondition post_acq(l)(st,st2) is nonempty.

acq_TCC1: OBLIGATION
  (FORALL (l: MName, st: (pre_acq(l))):
     nonempty?
       ({st2: sigma2 | post_acq(l)(st,st2)}));

This TCC has been proved interactively using PVS. The proof is unsurprising but not trivial: the user must supply a suitable candidate for the nondeterministic choice and then verify that all the various conditions imposed by the postŋ condition and the invariant are satis^ed.

We describe only the main highlights of the proof. Afŋ ter skolemizing, expanding de^nitions, and making some hidden hypotheses explicit, we are in a position where we may supply a possible candidate for the value that is nondeŋ terministically chosen. This is done using the tactic INST, which is given two arguments: an integer representing the appropriate subgoal and a value for instantiation consistŋ ing of a record representing the state after the operation is carried out. In this case, the postcondition of the acquire operation happens to be very explicit so the calculation of the post state simply re~ects the postcondition. Note that identi^ers such as st!1 are skolem variables generated by PVS.

(INST ŋ8
  "(# ms := union(ms(st!1), singleton(l!1)),
      b :=
      (: IF nth(b(st!1), 0) = Newest AND
            ms(st!1) = emptyset
         THEN Master ELSE nth(b(st!1), 0) ENDIF,
         IF nth(b(st!1), 1) = Newest AND
            ms(st!1) = emptyset
         THEN Master ELSE nth(b(st!1), 1) ENDIF,
         IF nth(b(st!1), 2) = Newest AND
            ms(st!1) = emptyset
         THEN Master ELSE nth(b(st!1), 2) ENDIF
      :) #)")

This gives us two subgoals: we must show that the witŋ ness satis^es both the postcondition and the invariant. We shall not describe these proofs in detail because they are lengthy and not particularly instructive. Brie~y, each subŋ goal consisted of a conjunction which was split to give several new subgoals. These subgoals were then proved by case analysis (splitting the hypotheses), rewriting and simplifying each individual case, and then using a decision procedure or a powerful tactic such as GRIND to either verify the conclusion or discover a contradiction within the hyŋ potheses. Much use was made of the HIDE command to hide irrelevant formulae and hence speed up the workings of the tactics.

Using a proof approach similar to that described above, we were able to show that the release operation is also feaŋ sible. The slave operation should have been similarly easy to handle, but, unfortunately, we were unable to complete the proof because of a bug in the PVS system concerning equality. (The PVS implementors are aware of this bug, and are working on correcting it.)

3.5. Some validation conditions

In our speci^cation of the slave operation, the postcondiŋ tions explicitly specify what happens to those buffers which have status Slave or Master, but do not describe the effect on buffers which have status Newest or Idle. However, one can show that, in conjunction with the invariant (and the frame in the VDM version), the postcondition ensures that no other Newest buffer remains, exactly one new Slave buffer is chosen, and no new Master buffers are added. This fact was stated as a validation condition in [6]. We have veri^ed it using PVS. We show its statement in both VDM and PVS notations.

∀ i ∈ {1, 2, 3} · ( ↼− b (i) ∈ {N, I} ⇒ b (i) ∈ {I, S})

slave_prop : CONJECTURE
   (FORALL (i:{x:nat| x=0 OR x=1 OR x=2}) :
      member(nth(b(st),i),
             {x:Status|x=Newest OR x=Idle})
      IMPLIES
      member(nth(b(slave(st)),i),
             {x:Status|x=Idle OR x=Slave}))

Our PVS proof of this statement required about 500 tactics and is dif^cult to follow intuitively, although neiŋ ther its structure nor any of the individual steps is particuŋ larly sophisticated. After some initial preparation including skolemization, de^nition expansion, and adding some simŋ ple lemmas, the tactic PROP, which does propositional simŋ pli^cation, is invoked. This has the effect of splitting the many disjunctions among the hypotheses and thereby breakŋ ing the top goal down into about 200 subgoals. The majority of these were proved easily by some de^nition expansion, simpli^cation, rewriting, and use of the tactic RECORD to deŋ tect contradictions among the hypotheses of the subgoal. Of the 54 subgoals which remained, 36 were proved by simply using GRIND. For the remaining subgoals, further case analŋ ysis was used to split each one into smaller subgoals which were then proved by GRIND.

The lemmas stated below were required later in order to prove a re^nement proof obligation. We present them as validation conditions since they are reasonable properties to require of the MSMIE system. Their proofs were carried out interactively and required about 15 tactics each.

slave_prop2 : LEMMA
  (FORALL (st,st2:sigma2) :
     post_slave(st,st2) IMPLIES
       count(Newest,b(st2)) = 1)

slave_prop3 : LEMMA
  (FORALL (st,st2:sigma2) :
     post_slave(st,st2) IMPLIES
      count(Master,b(st)) = count(Master,b(st2)))

4. A more abstract description: re�nement

The states of the MSMIE system may be described more abstractly by ignoring the identity of individual buffers and distinguishing only the possible combinations of buffers which satisfy the invariant. The two binary choices in the invariant concerning the number of buffers assigned to Masŋ ter and Newest mean that there are four such combinations: (Slave, Idle, Idle), (Slave, Idle, Newest), (Slave, Idle, Masŋ ter), and (Slave, Newest, Master).

4.1. The VDM speci�cation

In VDM, the possible status combinations are represented by giving a new enumerated type comprising four tokens.

types

Status1 = SII | SIN | SIM | SNM

The state simply records which combination is current and the invariant and initial state are the Şimages under retrievalŤ of the concrete ones given previously.

state Σ 1 of

bs : Status1 ms : MNameŋset

inv mkŋΣ 1 (bs, ms) 4

ms = {} ⇔ bs ∈ {SII, SIN}

init s 4 s = mkŋΣ 1 (SII, {}) end The operations are similar to those given in the previous speci^cations. In particular, the postconditions rely on the same case distinctions. As before, we do not show the speci^cation of the release operation.

operations

slave () ext wr bs : Status1

rd ms : (MNameŋset) pre true

post ( ↼− bs ∈ {SII, SIN} ⇒ bs = SIN) ∧ ( ↼− bs ∈ {SIM, SNM} ⇒ bs = SNM) ;

acq (l : MName) ext wr bs : Status1

wr ms : (MNameŋset) pre (¬ (l ∈ ms)) ∧ (¬ (bs = SII))

post ms = ↼−ms ∪ {l} ∧ if ↼−ms = {} then bs = SIM else bs = ↼− bs ;

4.2. The PVS speci�cation

We have formalized the more abstract speci^cation as a theory in PVS. The techniques used are the same as for the concrete speci^cation. The TCCs generated by typecheckŋ ing are also similar, but they are easier to prove because this is a less elaborate speci^cation.

msmie_sigma1[MName : TYPE+] : THEORY

BEGIN

Status1:TYPE = {SII,SIN,SIM,SNM}

sigma1:TYPE =
 {x : [# bs : Status1,
         ms : setof[MName] #] |
 (member(bs(x),{x:Status1|x=SII OR x=SIN})
  <=> ms(x) = emptyset)}

initial_sigma1:sigma1 =
  (# bs := SII, ms := emptyset #)

pre_slave:[sigma1 ŋ> bool] =
  LAMBDA (st:sigma1) : true

post_slave:[(pre_slave),sigma1 ŋ> bool] =
  LAMBDA (st:(pre_slave),st2:sigma1) :
    ((member(bs(st),{x:Status1|x=SII OR x=SIN})
         IMPLIES bs(st2) = SIN)
     AND
     (member(bs(st),{x:Status1|x=SIM OR x=SNM})
         IMPLIES bs(st2) = SNM))
     AND
     (ms(st2) = ms(st))

slave : [(pre_slave) ŋ> sigma1] =
  LAMBDA (st:(pre_slave)) :
    choose({st2:sigma1 | post_slave(st,st2)})

pre_acq:[MName ŋ> [sigma1 ŋ> bool]] =
  LAMBDA (l:MName)(st:sigma1) :
    (NOT member(l,ms(st))) AND (bs(st) /= SII)

post_acq:
  [l:MName ŋ> [(pre_acq(l)),sigma1 ŋ> bool]] =
  LAMBDA (l:MName)(st:(pre_acq(l)),st2:sigma1) :
    ((ms(st2) = union(ms(st),singleton(l)))) AND
     (IF ms(st) = emptyset
      THEN bs(st2) = SIM
      ELSE bs(st2) = bs(st)
      ENDIF)

acq:[l:MName, (pre_acq(l)) ŋ> sigma1] =
  LAMBDA (l:MName, st:(pre_acq(l))) :
    choose({st2:sigma1 | post_acq(l)(st,st2)})

4.3. The re�nement relationship

We can also formalize, as a PVS theory, the statement that sigma2 is a re^nement of sigma1. The re^nement relaŋ tionship is modeled by a theory which imports the theories representing the concrete and abstract speci^cations. The proof obligations that pertain to re^nement must be typed in by hand; they are not automatically generated by the system. They are declared to be CONJECTURES in the theory, as described below and then PVS is then used to prove them.

sigma1_sigma2[MName : TYPE+] : THEORY

BEGIN

IMPORTING msmie_sigma1[MName],msmie_sigma2[MName]

First, we de^ne the ŞretrieveŤ operation mapping conŋ crete states to abstract ones. The de^nition is given by cases, and is much the same as that given in [6]. Typeŋ checking generates a TCC stating that the retrieved state does indeed satisfy the invariant of sigma1. This is proved automatically by tcp.

retr2_1:[sigma2 ŋ> sigma1] = LAMBDA (st:sigma2) :
  (# bs :=
      LET cc : [nat,nat] =
        (count(Newest,b(st)),count(Master,b(st)))
      IN
        IF cc = (0,0) THEN SII
        ELSIF cc = (1,0) THEN SIN
        ELSIF cc = (0,1) THEN SIM
        ELSE SNM
        ENDIF,
     ms := ms(st) #)

Next we typed in the various proof obligations required to demonstrate that sigma2 is a re^nement of sigma1. All of these have been veri^ed using PVS. We shall indicate how dif^cult it was to carry out each veri^cation.

Showing adequacy was a relatively nonŋtrivial task, takŋ ing a few hours to complete. The proof itself is conceptually simple: we consider each of the possible values of the sysŋ tem status in the abstract speci^cation, and supply a suitable value for the concrete state in each case. For each of these we must then show two things: that it is the value returned by the retrieve function, and that it satis^es the invariant of sigma2. The ^rst can be proved almost automatically by expanding some de^nitions and invoking the tactic GRIND. To prove the second we must use the invariant of the abstract state. The entire proof script for adequacy is about 130 lines long.

adeq2_1 : CONJECTURE
  (FORALL (st1:sigma1) :
     (EXISTS (st2:sigma2) : retr2_1(st2) = st1))

Next we proved that the retrieve function maps the conŋ crete initial state to the abstract initial state. This was very simple: after expanding one de^nition, the GRIND tactic completed the proof.

init2_1 : CONJECTURE
  (retr2_1 (initial_sigma2) = initial_sigma1)

Next was the domain rule for the slave operation. This is trivial and was proved automatically by GRIND.

slave_dom2_1 : CONJECTURE
  (FORALL (st2:sigma2) :
      (pre_slave (retr2_1(st2))) IMPLIES
         pre_slave(st2))

To prove the result rule for slave we ^rst added the lemŋ mas slave_prop1 and slave_prop2 which were validation conditions in the theory msmie_sigma2. Next, we used the tactic TYPEPRED to make the invariant of sigma2 visible to the prover. Once this preparation was in place, the GRIND tactic completed the proof.

slave_result2_1 : CONJECTURE
 (FORALL
   (st2_:((pre_slave:[sigma2ŋ>bool])),st2:sigma2):
   ((pre_slave (retr2_1(st2_)) AND
    (post_slave(st2_,st2))) IMPLIES
      (post_slave (retr2_1(st2_),retr2_1(st2)))))

The domain rule for acq required a proof about 30 tactics long as well as a lemma about the count function. The proof structure is as follows: ^rst de^nitions were expanded, the lemma was added, hidden type information was made visible, and the proof was split into two subgoals; then the

GRIND tactic was called upon to complete the proof.

acq_dom2_1 : CONJECTURE
  (FORALL (st2:sigma2) : (FORALL (l:MName) :
     (pre_acq (l)(retr2_1(st2))) IMPLIES
        pre_acq(l)(st2)))

The result rule for acq required a complicated, interactive proof comprising hundreds of tactics which took several days to complete.

acq_result2_1 : CONJECTURE
  (FORALL (l:MName) :
  (FORALL
     (st2_:(pre_acq(l):[sigma2ŋ>bool]),st2:sigma2):
     ((pre_acq(l)(retr2_1(st2_)) AND
      (post_acq(l)(st2_,st2)))
      IMPLIES
       (post_acq(l)(retr2_1(st2_),retr2_1(st2)))
       )))

The domain rule (not shown) for the release operation was proved automatically by GRIND. The result rule for this operation required a complex, interactive proof very similar to that of the result rule for acq.

5. Results

Generally, the combination of the PVS speci^cation lanŋ guage with VDM speci^cation and re^nement style seems to work well. PVS can express a large, useful subset of VDMŋ SL, despite its lack of union types and its restriction to total functions only. Indeed, much, but not all, of the expresŋ siveness of partial functions in VDMŋSL can be captured in PVS by the use of subtyping and dependent types. Some examples of partial functions which show this are given beŋ low in Section 5.1. Section 5.2 discusses the treatment of looseness in VDM and PVS.

We were impressed by the fact that we were actually able to prove all (but one) of the proof obligations, including re^nement proof obligations, for this development using PVS. This augers well for the usability of the system for further applications.

Like [1], we made use of the TCC mechanism of PVS to obtain some automatic proof obligation generation. This results in TCCs which are, in general, too complicated to be solved by the PVS command ŞtypecheckŋproveŤ, which is good at automatically ^nding proofs of simple TCCs. Unfortunately, in the present implementation of PVS it is impossible to prevent this command from embarking on timeŋconsuming attempts to prove all TCCs, even though the user may know that certain ones are too dif^cult to be solved. A more ~exible version of ŞtypecheckŋproveŤ, or perhaps simply a time limit to its operation, would be welcome.

Translating from VDM to PVS must be done by hand. There is also no automatic proof obligation generation for proof obligations arising from re^nement. The closed deŋ sign of the PVS system means that it may be dif^cult to implement extensions to it which would support these proŋ cesses. In practice, there is no need for the translation if the user is willing to work directly with the PVS speci^cation language. The lack of automatic translation would make it a timeŋconsuming process to apply our approach to translation to the large body of existing VDM speci^cations. However, the time to translate and individual speci^cation is small relŋ ative to the time taken to prove it and so this lack of support for this should not deter the serious formal developer.

5.1. Partiality in VDM and PVS

The most notable difference between the speci^cation languages of PVS and VDM is that PVS deals only in total functions. In practice, much of the language ~exibility of partial functions in LPF can be captured in PVS by the use of subtypes and dependent types to express the domain of de^nition of a function. A good example is the nth function on lists which is de^ned recursively in PVS as follows:

nth(l,(n:nat | n < length(l))): RECURSIVE nat =
  IF n=0 THEN car(l) ELSE nth(cdr(l),nŋ1) ENDIF
  MEASURE length(l)

As the following examples show, this function may be used freely when writing speci^cations: the fact that it is partial imposes no special syntactic constraints. For examŋ ple we can write:

ex1:nat = nth((: 1, 3 , 2 :), 1)

ex2:nat = nth((: 1, 3 , 2 :), 4)

Correctness is maintained by typechecking. The ^rst example causes no problems. However, the second example results in this false TCC:

4 < length[nat]((: 1, 3, 2 :))

An example of a use of partial functions which is possible in LPF but not in PVS involves the subp function, due to Cliff Jones. In PVS it may be de^ned as follows^2.

2This de^nition is due to Klaus Havelund.

subp(i:nat,j:nat | i >= j): RECURSIVE nat =
  IF i=j THEN 0 ELSE 1 + subp(i,j+1) ENDIF
  MEASURE abs(iŋj)

When applied to natural numbers i and j, where i > j, this function returns the difference iŋj. This property can be formalized and proved in both PVS and LPF. However, the following property which is also true of subp in LPF, cannot be proved in PVS Ů in fact, it cannot even be typechecked.

subp_lemma : CONJECTURE
  FORALL (i,j:nat) :
    (subp(i,j) = iŋj) OR (subp(j,i) = jŋi)

Attempting to typecheck this results in the false TCC:

subp_lemma_TCC1: OBLIGATION
  (FORALL (i, j: nat): i >= j);

Fortunately, the present example does not contain any construction where this distinction is signi^cant.

5.2. Looseness in VDM and PVS

Another semantic difference between VDM and our translation to PVS is in the interpretation of expressions whose values are not fully determined.

In VDM, looseness in function de^nitions is interpreted as underspeci^cation, that is to say, every invocation of a function with the same argument will return the same result; whereas looseness in operation de^nitions is understood to be genuine nonŋdeterminism, so separate invocations of a loosely speci^ed operation can yield different results even if called with the same arguments and in the same state[11]. The motivation for this distinction is that, in an implemenŋ tation, the result of an operation may depend on some state not being modeled in the abstraction, whereas a function should be declarative however it is implemented.

For example, for functions f and g:

f (x ) = let y be st y > x in y end g(x ) = let y be st y > x in y end

we can always be certain that f (x ) = f (x ), but f (x ) = g(x ) may not necessarily hold in a re^nement.

PVS, on the other hand, takes a more constrained interŋ pretation of looseness: all occurrences of the same choice expression must yield the same result wherever they occur. So, if we make the corresponding de^nitions in PVS,

f(x) = choose({y:nat | y > x})
g(x) = choose({y:nat | y > x})

then we always have f(x) = g(x) also.

The VDM interpretation of loose functions is appropriŋ ate in the context of a development employing the Şdesign by contractŤ paradigm. Underspeci^cation represents the deferral of a design decision concerning the choice of a fully determined implementation. Thus looseness can be removed during re^nement and the resulting behavior will be no worse from the caller’s point of view than that of the loose function. However, this interpretation of looseness has severe implications for reasoning as it prohibits the inŋ discriminate substitution of equals. We cannot make the simple chain of equalities:

f (x ) = let y be st y > x in y end = g(x )

Rather, each occurrence of a loose expression must in some way be tagged in order that it is possible to determine which occurrence is being referred to when it occurs in proofs. It also means that betaŋreduction can only be undertaken when the argument is fullyŋdetermined[11].

The PVS interpretation of looseness yields simpler proofs since we can be sure that identical expressions will have equal value irrespective of how they arise. However, this interpretation of looseness de^es compositionality in re^neŋ ment as if the same choice expression occurs in two separate parts of a speci^cation, they must both be treated similarly in any subsequent re^nement.

In the present example,we have interpreted implicit operŋ ations by use of the PVS choice operator. In the case where there is some genuine choice, as in slave, this is not strictly correct. With this interpretation we could prove properties of the speci^cation which are not necessarily preserved by an implementation.

These two differences in the semantics raise methodologŋ ical questions about the use of such constructions in practice. Though partiality and looseness are both extremely useful, they should be used with caution particularly in circumŋ stances where there is disagreement as to their interpretaŋ tion.

6. Conclusion

The VDMŋSL style of speci^cation and re^nement ^ts well with the PVS language. As a result we are able to use a very direct embedding of VDM in PVS (for example, logical formulae in VDM are represented by those of PVS), which means that the proof capabilities of the PVS system are available directly to the VDM user. This is not always the case where a deeper, more indirect embedding is required, forcing the user to navigate through layers of de^nitions.

The extensive typeŋchecking done by the PVS system contributes greatly to our con^dence in the correctness of speci^cations and re^nements. For example, we can be certain that functions are applied only to arguments within their domain. We also derived con^dence from the proof process. Although this is somewhat muted by reports of bugs in the PVS prover, the advantages of mechanical supŋ port for proof compared to making proofs by hand almost certainly outweighs the possibility of the system constructŋ ing an erroneous proof.

At present the translation from VDMŋSL to PVS, and the generation of re^nement proof obligations must both be done manually. As well as being inconvenient, these manual processes are opportunities for the introduction of errors. It is possible to automate the translation step outside of PVS, generating PVS theories from VDMŋSL speci^caŋ tions. However, the closed nature of the PVS system makes it dif^cult to achieve a close integration with other tools supporting VDMŋSL such as can be achieved with other more open systems [3, 2]. This leads to the disadvantage that a VDMŋSL user who wishes to use PVS for proofs must master the PVS notation and become, in effect, a PVS user as well.

The PVS prover was suf^ciently fast and powerful to make it feasible to do proofs of the size attempted in this example, though the time required to do this is considerŋ able and so proof for ŞrealŋworldŤ applications remains an expensive activity. On the other hand, the proofs which were undertaken, though not mathematically sophisticated, involved such elaborate case analyses that it is unlikely that they would be successfully carried out without the help of tools.

Further work needs to be carried out to discover the imŋ plications, as far as re^nement is concerned, of the differŋ ent approaches to looseness taken in VDMŋSL and PVS. It would also be interesting to see how well the approach scales up to larger, more ŞrealŋworldŤ applications which might provide a more exacting test of the capabilities of the PVS system. Finally, the authors are interested in a comparŋ ison between PVS and the new VDM proof tool based on Isabelle which is currently under development at IFAD.

References

[1] S. Agerholm. Translating Speci^cations in VDMŋSL to PVS, Proceedings of the 9th Inŋ ternational Conference on Theorem Proving in Higher Order Logics (TPHOL ’96) LNCS 1125, Springer, 1996

[2] S. Agerholm and J. Frost. An Isabelleŋbased Theorem Prover for VDMŋSL. In Proceedings of the 10th International Conference on Theorem Proving in Higher Order Logics (TPHOLs’97), SpringerŋVerlag LNCS, 1997.

[3] S. Agerholm and J. Frost. Towards an Integrated CASE and Theorem Proving Tool for VDMŋSL. In FME’97, SpringerŋVerlag LNCS, 1997.

[4] D.J. Andrews and H. Bruun et al. Information Technology Ů Programming Languages, their environments and system software interfaces Ů Vienna Development MethodŋSpeci^cation Lanŋ guage Part 1: Base language ISO Draft Internaŋ tional Standard: 13817Ű1, 1995

[5] G. Bruns and S. Anderson. The Formalizaŋ tion and Analysis of a Communications Protoŋ col, Formal Aspects of Computingm Volume 6, Number 1, Springer, 1994

[6] J.C. Bicarregui. A ModelŋOriented Analysis of a Communications Protocol, Technical report RALŋ93ŋ099, Rutherford Appleton Laboratory, 1993.

[7] J.C. Bicarregui and B. Ritchie. Invariants, Frames and Postconditions: a comparison of the VDM and B notations, Proceeding of Forŋ mal Methods Europe ’93, Woodcock and Larsen (Eds), LNCS 670, SpringerŋVerlag. Also in IEEE Transaction on Software Engineering, Vol. 21, No. 2, pp. 79ŋ89, Feb. 1995.

[8] J. Crow, S. Owre, J. Rushby, N. Shankar and M. Srivas. A Tutorial Introduction to PVS, Comŋ puter Science Laboratory, SRI International, Menlo Park, CA, February 1993.

[9] R. Elmstrøm, P.G. Larsen, and P.B. Lassen. The IFAD VDMŋSL Toolbox: A practical approach to formal speci^cations. ACM Sigplan Notices, 29(9):77ŋ80, September 1994.

[10] C.B. Jones. Systematic Software Development using VDM (second edition), Prentice Hall, 1990.

[11] Peter G. Larsen, Bo S. Hansen. Semantics of Underdetermined Expressions, Formal Aspects of Computing, vol. 8, no. 1, 1996.

[12] C.H. Pratten, An Introduction to proving AMN Speci^cations with PVS and the AMNŋProof Tool, Z Twenty Years onŋ What is its Future?, Proceedings of the 7th International Conference on: Putting into practice methods and tools for information systems design. H. Habrias (ed), ISBN: 2ŋ906082ŋ19ŋ8, Nantes, France, October 1995.

[13] L.L. Santoline et al. Multiprocessor Sharedŋ Memory Information Exchange, IEEE Transacŋ tions on Nuclear Science. Vol. 36. No. 1, Februŋ ary 1989, pp. 626ŋ633.